

Appendix

MPLUS Setup Files for Models in the Present Investigation

```

TITLE: Model 0 Single-level model with manifest L1 and L2 variables;
DATA: File is bflpe.dat;
VARIABLE: Names are
  idstud idsch sex szweig1
  zmsc1 zmsc2 zmsc3 zmsc4
  zmach1 zmach2 zmach3
  zmsct zmacht
  s_msc1 s_msc2 s_msc3 s_msc4 s_msct
  s_mach1 s_mach2 s_mach3 s_macht;
USEVAR ARE
  zmsct s_macht gr_math ;

DEFINE:
  gr_math = zmacht - s_macht;
ANALYSIS: Type is general ;
MODEL:
  zmsct on gr_math (ind_ach);
  zmsct on s_macht (Sch_Ach);

MODEL CONSTRAINT:
  new(bflpe);
  bflpe = Sch_Ach - ind_ach;

OUTPUT: sampstat stand tech1;

```

```

TITLE: MODEL 1 Doublely Manifest Multilevel Model
with manifest constructs and manifest aggregation;
DATA: File is bflpe.dat;
VARIABLE: Names are
  idstud idsch sex szweig1
  zmsc1 zmsc2 zmsc3 zmsc4
  zmach1 zmach2 zmach3
  zmsct zmacht
  s_msc1 s_msc2 s_msc3 s_msc4 s_msct
  s_mach1 s_mach2 s_mach3 s_macht;
USEVAR ARE
  zmsct zmacht s_macht;
  cluster = idsch;
  within = zmacht ;
  between = s_macht;
  centering = groupmean(zmacht);
  missing are all (-99);
ANALYSIS: Type is twolevel ;
MODEL:

  %within%
  zmsct on zmacht (b_within);

  %between%
  zmsct on s_macht (b_betwn);

MODEL CONSTRAINT:
  new(bflpe);
  bflpe = b_betwn - b_within;
OUTPUT: sampstat stand tech1;

```

```

TITLE: model 3 Latent-manifest model with Latent constructs
based on multiple indicators, but manifest aggregation;
DATA: File is bflpe.dat;
VARIABLE: Names are
  idstud idsch sex szweig1
  zmsc1 zmsc2 zmsc3 zmsc4
  zmach1 zmach2 zmach3
  zmsct zmacht
  s_msc1 s_msc2 s_msc3 s_msc4 s_msct
  s_mach1 s_mach2 s_mach3 s_macht;
USEVAR ARE
  zmsc1 zmsc2 zmsc3
  zmsc4 zmach1 zmach2 zmach3
  s_mach1 s_mach2 s_mach3;
  cluster = idsch;
  within = zmach1 zmach2 zmach3;

```

```

between = s_mach1 s_mach2 s_mach3;
centering = groupmean(zmach1 zmach2 zmach3);
missing are all (-99);
ANALYSIS: Type is twolevel ; algorithm=em; mconv=1000;
ANALYSIS: Type is twolevel ; algorithm=em; mconv=1000;
MODEL:
%within%
MSC_W by zmsc1 (1) ;
MSC_W by zmsc2 (2);
MSC_W by zmsc3 (3) ;
MSC_W by zmsc4 (4);

MACH_W by zmach1 (5) ;
MACH_W by zmach2 (6);
MACH_W by zmach3 (7) ;

MSC_W on MACH_W (b_within);

%between%
MSC_B by zmsc1 (1) ;
MSC_B by zmsc2 (2);
MSC_B by zmsc3 (3) ;
MSC_B by zmsc4 (4);

MACH_B by s_mach1 (5) ;
MACH_B by s_mach2 (6);
MACH_B by s_mach3 (7) ;

MSC_B on MACH_B (b_between);

MODEL CONSTRAINT:
new(bflpe);
bflpe = b_between - b_within;

OUTPUT: sampstat stand tech1;

TITLE: Model 3 Manifest-Latent Model with manifest Single-indicators
of each construct, but latent aggregation from L1 to L2.;
DATA: File is bflpe.dat;
VARIABLE: Names are
idstud idsch sex szweig1
zmsc1 zmsc2 zmsc3 zmsc4
zmach1 zmach2 zmach3
zmsct zmacht
s_msc1 s_msc2 s_msc3 s_msc4 s_msct
s_mach1 s_mach2 s_mach3 s_macht;
USEVAR ARE
zmsct zmacht;
cluster = idsch;

ANALYSIS: Type is twolevel ;
MODEL:

%within%
zmsct on zmacht (b_within);

%between%
zmsct on zmacht (b_betwn);

MODEL CONSTRAINT:

new(bflpe);
bflpe = b_betwn - b_within;
OUTPUT: sampstat stand tech1;

TITLE: Model 4 Doubly-latent MLM with latent constructs
based on multiple indicators and latent aggregation;
mult indicators; no rand slopes;
DATA: File is bflpe.dat;
VARIABLE: Names are
idstud idsch sex szweig1
zmsc1 zmsc2 zmsc3 zmsc4
zmach1 zmach2 zmach3
zmsct zmacht

```

```

s_msc1 s_msc2 s_msc3 s_msc4 s_msct
s_mach1 s_mach2 s_mach3 s_macht;
USEVAR ARE
zmsc1 zmsc2 zmsc3
zmsc4 zmach1 zmach2 zmach3;
cluster = idsch;
centering = grandmean(zmach1 zmach2 zmach3);
missing are all (-99);
ANALYSIS: Type is twolevel ; algorithm=em; mconv=1000;
MODEL:
%within%
MSC_W by zmsc1 (1);
MSC_W by zmsc2 (2);
MSC_W by zmsc3 (3);
MSC_W by zmsc4 (4);
MACH_W by zmach1 (5);
MACH_W by zmach2 (6);
MACH_W by zmach3 (7);
MSC_W on MACH_W (b_within);

%between%
MSC_B by zmsc1 (1);
MSC_B by zmsc2 (2);
MSC_B by zmsc3 (3);
MSC_B by zmsc4 (4);
MACH_B by zmach1 (5) ;
MACH_B by zmach2 (6);
MACH_B by zmach3 (7);

MSC_B on MACH_B (b_betwn);

MODEL CONSTRAINT:
new(bflpe);
bflpe = b_betwn - b_within;
OUTPUT: sampstat stand tech1;

TITLE: Model 5A Doubly-latent MLM (Model 5) with random slopes
DATA: File is bflpe.dat;
VARIABLE: Names are
idstud idsch sex szweig1
zmsc1 zmsc2 zmsc3 zmsc4
zmach1 zmach2 zmach3
zmsct zmacht
s_msc1 s_msc2 s_msc3 s_msc4 s_msct
s_mach1 s_mach2 s_mach3 s_macht;
USEVAR ARE
zmsc1 zmsc2 zmsc3
zmsc4 zmach1 zmach2 zmach3;
cluster = idsch;
missing are all (-99);
ANALYSIS:Type is twolevel random; algorithm=integration; integration=10;
GHFIML=OFF;

%within%
MODEL:
%within%
MSC_W by zmsc1 (1);
MSC_W by zmsc2 (2);
MSC_W by zmsc3 (3);
MSC_W by zmsc4 (4);
MACH_W by zmach1 (5);
MACH_W by zmach2 (6);
MACH_W by zmach3 (7);

s | MSC_W on MACH_W;

%between%
MSC_B by zmsc1 (1);
MSC_B by zmsc2 (2);
MSC_B by zmsc3 (3);
MSC_B by zmsc4 (4);
MACH_B by zmach1 (5) ;
MACH_B by zmach2 (6);
MACH_B by zmach3 (7);

[s] (b_within);

```

```

MSC_B on MACH_B (b_betwn);

MODEL CONSTRAINT:

    new(bflpe);
    bflpe = b_betwn - b_within;
OUTPUT: sampstat tech1; !stand

    TITLE: Model 5B Doubly-latent MLM (Model 4) with random slopes and
Cross-level latent interaction (s on MACH_B);
DATA: File is bflpe.dat;
VARIABLE: Names are
    idstud idsch sex szweig1
    zmsc1 zmsc2 zmsc3 zmsc4
    zmach1 zmach2 zmach3
    zmsct zmacht
    s_msc1 s_msc2 s_msc3 s_msc4 s_msct
    s_mach1 s_mach2 s_mach3 s_macht;
USEVAR ARE
    zmsc1 zmsc2 zmsc3
    zmsc4 zmach1 zmach2 zmach3;
cluster = idsch;
missing are all (-99);
ANALYSIS:Type is twolevel random; algorithm=integration; integration=10;
    GHFIML=OFF;

    %within%
MODEL:
    %within%
    MSC_W by zmsc1 (1);
    MSC_W by zmsc2 (2);
    MSC_W by zmsc3 (3);
    MSC_W by zmsc4 (4);
    MACH_W by zmach1 (5);
    MACH_W by zmach2 (6);
    MACH_W by zmach3 (7);

    s | MSC_W on MACH_W;

    %between%
    MSC_B by zmsc1 (1);
    MSC_B by zmsc2 (2);
    MSC_B by zmsc3 (3);
    MSC_B by zmsc4 (4);
    MACH_B by zmach1 (5);
    MACH_B by zmach2 (6);
    MACH_B by zmach3 (7);
    s on MACH_B;
    [s] (b_within);

    MSC_B on MACH_B (b_betwn);

MODEL CONSTRAINT:

    new(bflpe);
    bflpe = b_betwn - b_within;
OUTPUT: sampstat tech1; !stand

TITLE: Model 6 Doubly-latent model (Model 4) with the addition
Of L1-Gender and its latent interaction with L1-achievement;
DATA: File is bflpe.dat;
VARIABLE: Names are
    idstud idsch sex szweig1
    zmsc1 zmsc2 zmsc3 zmsc4
    zmach1 zmach2 zmach3
    zmsct zmacht
    s_msc1 s_msc2 s_msc3 s_msc4 s_msct
    s_mach1 s_mach2 s_mach3 s_macht;
USEVAR ARE
    zmsc1 zmsc2 zmsc3 sex
    zmsc4 zmach1 zmach2 zmach3;
cluster = idsch;
WITHIN ARE sex;
ANALYSIS: Type is twolevel random; algorithm=integration; integration=10;
    GHFIML=OFF;
MODEL:

```

```

%within%
MSC_W by zmsc1 (1);
MSC_W by zmsc2 (2);
MSC_W by zmsc3 (3);
MSC_W by zmsc4 (4);
MACH_W by zmach1 (5);
MACH_W by zmach2 (6);
MACH_W by zmach3 (7);
ISxXIMac | MACH_W xwith sex;
MSC_W on MACH_W sex ISxXIMac;
MSC_W on MACH_W (b_within);
%between%
MSC_B by zmsc1 (1);
MSC_B by zmsc2 (2);
MSC_B by zmsc3 (3);
MSC_B by zmsc4 (4);
MACH_B by zmach1 (5);
MACH_B by zmach2 (6);
MACH_B by zmach3 (7);
MSC_B on MACH_B (b_betwn);
MODEL CONSTRAINT:
new(bflpe);
bflpe = b_betwn - b_within;
OUTPUT: sampstat stand tech1;

```

TITLE: Model 7 Doubly-latent model (Model 4) with the addition
Of a latent quadratic component of L1-achievement;

DATA: File is bflpe.dat;

VARIABLE: Names are

```

idstud idsch sex szweig1
zmsc1 zmsc2 zmsc3 zmsc4
zmach1 zmach2 zmach3
zmsct zmacht
s_msc1 s_msc2 s_msc3 s_msc4 s_msct
s_mach1 s_mach2 s_mach3 s_macht;

```

USEVAR ARE

```

zmsc1 zmsc2 zmsc3
zmsc4 zmach1 zmach2 zmach3;
cluster = idsch;
missing are all (-99);

```

ANALYSIS:Type is twolevel random;

algorithm=integration em;integration=5;mconv=1000; GHFIML=OFF;

MODEL:

```

%within%
MSC_W by zmsc1 (1);
MSC_W by zmsc2 (2);
MSC_W by zmsc3 (3);
MSC_W by zmsc4 (4);
MACH_W by zmach1 (5);
MACH_W by zmach2 (6);
MACH_W by zmach3 (7);
L1MACSQ | MACH_W xwith MACH_W;
MSC_W on MACH_W L1MACSQ;
MSC_W on MACH_W (b_within);
%between%
MSC_B by zmsc1 (1);
MSC_B by zmsc2 (2);
MSC_B by zmsc3 (3);
MSC_B by zmsc4 (4);
MACH_B by zmach1 (5);
MACH_B by zmach2 (6);
MACH_B by zmach3 (7);
MSC_B on MACH_B (b_betwn);
MODEL CONSTRAINT:
new(bflpe);
bflpe = b_betwn - b_within;
OUTPUT: sampstat stand tech1;

```

TITLE: Model 8 (Model 4 with alternative parameterizations of effect size);

DATA: File is bflpe.dat;

VARIABLE: Names are

```

idstud idsch sex szweig1
zmsc1 zmsc2 zmsc3 zmsc4
zmach1 zmach2 zmach3
zmsct zmacht

```

```

s_msc1 s_msc2 s_msc3 s_msc4 s_msct
  s_mach1 s_mach2 s_mach3 s_macht;
USEVARIABLES ARE
  zmsc1 zmsc2 zmsc3
  zmsc4 zmach1 zmach2 zmach3;
cluster = idsch;
ANALYSIS: Type is twolevel ; algorithm=em; mconv=1000; GHFIML=OFF;
MODEL:
  %within%
  MSC_W by zmsc1 (1);
  MSC_W by zmsc2 (2);
  MSC_W by zmsc3 (3);
  MSC_W by zmsc4 (4);
  MACH_W by zmach1 (5);
  MACH_W by zmach2 (6);
  MACH_W by zmach3 (7);
  MSC_W on MACH_W (b_within);
  MACH_W (Psi_W);
  MSC_W (Theta_W);
  %between%
  MSC_B by zmsc1 (1);
  MSC_B by zmsc2 (2);
  MSC_B by zmsc3 (3);
  MSC_B by zmsc4 (4);
  MACH_B by zmach1 (5) ;
  MACH_B by zmach2 (6);
  MACH_B by zmach3 (7);
  MACH_B (Psi_B);
  MSC_B (Theta_B);
  MSC_B on MACH_B (b_betwn);
MODEL CONSTRAINT:
  new(bflpe);
  bflpe = b_betwn - b_within;
  new(st-bflpe);
  st-bflpe = bflpe*(sqrt(Psi_B)
  /sqrt(Psi_B*b_betwn**2 + Theta_B + Psi_W*b_within**2 + Theta_W));
  new(ES1);
  ES1 = bflpe*(2*sqrt(Psi_B) /sqrt(Theta_W));
  new(ES2);
  ES2= bflpe*(2*sqrt(Psi_B)/sqrt(Psi_W*b_within**2 + Theta_W));
  new(ES3);
  ES3= bflpe*(2*sqrt(Psi_B)/sqrt((Psi_B*b_betwn**2 + Psi_W*b_within**2 + Theta_W));
OUTPUT: residual sampstat stand tech1 MODINDICES tech4 ;

```